# $Id$
# Maintainer: Ionut Biru <ibiru@archlinux.org>
# Contributor: Timm Preetz <timm@preetz.us>
pkgname=telepathy-haze
pkgver=0.8.0
pkgrel=1
pkgdesc="A telepathy-backend to use libpurple (Pidgin) protocols."
arch=('i686' 'x86_64')
url="http://telepathy.freedesktop.org"
license=('GPL')
depends=('telepathy-glib' 'libpurple')
makedepends=('libxslt' 'python2')
replaces=('telepathy-butterfly')
install=telepathy-haze.install
groups=('telepathy')
source=(https://telepathy.freedesktop.org/releases/$pkgname/$pkgname-$pkgver.tar.gz{,.asc})
md5sums=('b9ee3638833fb50db6276d1b771820b0'
         'SKIP')
validpgpkeys=(DA98F25C0871C49A59EAFF2C4DE8FF2A63C7CC90) # Simon McVittie <smcv@pseudorandom.co.uk>

build() {
  cd "$pkgname-$pkgver"
  PYTHON=/usr/bin/python2 ./configure --prefix=/usr \
      --libexecdir=/usr/lib/telepathy
  make
}

package(){
  cd "$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
